

# A Configuration File Companion: testing and using environment variables and options; templates for startup-only options `initstmt` and `termstmt`

Ronald J. Fehd, senior maverick, theoretical programmer,  
Fragile-Free Software Institute

**Abstract**

**description** The startup process of SAS® software reads one or more configuration files, \*.cfg, which have allocations of environment variables, the values of which are used in SAS startup-only options to provide access to libraries, lists of directories that contain files that SAS uses for functions, macros, and procedures.

**purpose** This paper provides programmers and advanced users programs to review the default configuration files; procedures, options, and sql to discover options; and a suite of programs to use in Test-Driven Development (TDD) to trace and verify user-written configuration files.

**audience** developers, programmers, advanced users

**keywords** API: FOLDERID\_Documents;  
CSIDL Equivalents: CSIDL\_MYDOCUMENTS, CSIDL\_PERSONAL  
sql dictionary.options, opstart='startup';  
environment variables: mysasfiles, path, sasautos, sascfg, sasroot;  
batch files: sas.cmd, 01-sas-v9-cfg-test.\*;  
startup-only options: echoauto, initstmt, sysin, termstmt, verbose

---

<b>In this paper:</b>	<b>Bricolage: tools for discovery</b>	<b>3</b>
	values of opstart . . . . .	4
	echoing configuration files . . . . .	5
	sql procedure, dictionary.options . . . . .	6
	<b>Setup for testing</b>	<b>7</b>
	unit test: sasv9.cfg . . . . .	10
	template for initstmt and termstmt . . . . .	13
	unit test: autoexec.sas . . . . .	15
	<b>Summary</b>	<b>17</b>
	<b>References</b>	<b>18</b>

---

<b>List of tables:</b>	1	startup process . . . . .	3
	2	frequency of optstart . . . . .	3

---

<b>List of programs:</b>	1	proc-options-define-value-config.sas . . . . .	4
	2	echo-any-file.sas . . . . .	5
	3	echo-config-primary.sas . . . . .	5
	4	echo-config-main.sas . . . . .	6
	5	proc-sql-describe-dictionary-options.sas . . . . .	6
	6	proc freq, tables optstart . . . . .	6
	7	write test suite of config test files . . . . .	8
	8	project sas.cmd . . . . .	9
	9	project sasv9.cfg . . . . .	9
	10	01-config-test.* . . . . .	10
	11	initstmt.sas . . . . .	14
	12	termstmt.sas . . . . .	14
	13	02-autoexec-test: .bat and .sas . . . . .	15

---

## Introduction

### overview

This is the overview, which consists of a list of topics in this section.

- what's inside configuration files
- startup process
- why use configuration files

### what's inside

Configuration files contain two types of statements: allocation of environment variables and assignment of options. The main configuration file contains over 200 lines; more than half of those lines contain references to the environment variable `sasroot` in environment variables' and options' assignments.

---

## startup process

The startup process of SAS software has two insertion points for startup-only options: on the command line, and in configuration files. Table 1 lists the insertion points and the startup-only options.

**Table 1 startup process**

1. command-line:				<code>sas -sysin job-name &lt;-options&gt;</code>
2. configuration files:				
(a) primary:				<code>!sasroot/sasv9.cfg</code>
(b) main:	required			<code>!sasroot/nls/??/sasv9.cfg</code>
(c) tertiary:	project:			<code>sasv9.cfg</code>
3. options from command-line				
code-insertion points: startup-only options				
option				
<u>name</u>	<u>default</u>	<u>where set</u>		<u>assignment</u>
config	sasv9.cfg			<code>-config 'filename.ext'</code>
autoexec	autoexec.sas			<code>-autoexec 'filename.ext'</code>
initstmt		command-line or config		<code>-initstmt '*text;'</code>
sysparm		command line:		<code>-sysparm "text"</code>
sysin		command line:		<code>-sysin job-name</code>
sysparm		macro variable, in program		<code>%let sysparm = value;</code>
termstmt		command-line or config		<code>-termstmt '*text;'</code>

**notes:** Description of this process is found in SAS Documentation under the topic: Files Used by SAS and Customizing Your SAS Session by Using Configuration and Autoexec Files. Your installation may have several folders under National Language Support (nls); English (en) is shown here. See also the option `locale`.

## why use config?

Table 2 shows that almost one third of options are startup-only.

**Table 2 frequency of optstart**

optstart	Option Set		Cumulative Frequency	Cumulative Percent
	Frequency	Percent		
anytime	315	67.74	315	67.74
startup	150	32.26	465	100.00

Program References `config:prg:freq-optstart`, pg. 6 produces this frequency table.

## Bricolage: tools for discovery

### overview

This is the list of topics in this section.

- values of optstart
- echoing configuration files
- sql procedure, dictionary.options

## values of opstart

There are two values of opstart: anytime and startup.  
In the output of program proc-options-define-value-config.sas these values are seen in the output table in the log in the section When Can Set:

When Can Set: Startup or anytime during the SAS Session

When Can Set: Session startup (command line or config) only

---

### Program 1 proc-options-define-value-config.sas

```
proc options define value option=config;
run;
```

---

#### log

see line 23, for optstart='startup'

```
1 Option Value Information For SAS Option CONFIG
2 Value: ( "C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg"
3         "C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"
4         "C:\SAS-projects\2023-config\sas-startup-configuration\sasv9.cfg"
5         )
6 Scope: SAS Session
7 How option value set: Internal
8
9 Option Definition Information for SAS Option CONFIG
10 Group= INSTALL
11 Group Description: Site installation settings
12 Description: Specifies the configuration file that is used
13 when initializing or overriding the values of SAS system options.
14 Type: The option value is of type CHARACTER
15 Maximum Number of Characters: 32000
16 Casing: The option value is retained with original casing
17 Quotes: If present during "set", start and end quotes are removed
18 Parentheses: The option value does not require enclosure
19 within parentheses.
20 If present, the parentheses are retained.
21 Expansion: Environment variables, within the option value,
22 are not expanded
23 When Can Set: Session startup (command line or config) only
```

---

## echoing configuration files

### overview

This is the overview, which consists of a list of programs in this section.

- config syntax
- echo any file
- echo config primary
- echo config main

### config syntax

While viewing any configuration file, \*.cfg, look for these two statements:

- allocate environment variable: `-set evar value`
- assign option: `-<option-name> value`

Program 2 is a subroutine used by the programs 3: echo-config-primary.sas and 4: echo-config-main.sas.

#### Program 2 echo-any-file.sas

```
%put echo &=in_file;
document
%let lrecl = 80;
data _null_;
    infile "&in_file"  lrecl = &lrecl  pad
                                end    = endofile;

do until(endofile);
    input @1 line $char&lrecl.;
    put   @1 line $char&lrecl.;
end;

stop;
run;
```

Program 3 echos the primary configuration file in the sasroot directory.

#### Program 3 echo-config-primary.sas

```
%put echo &=in_file;
echo-config-primary
%let lrecl = 80;
data _null_;
    infile "&in_file"  lrecl = &lrecl  pad
                                end    = endofile;

do until(endofile);
    input @1 line $char&lrecl.;
    put   @1 line $char&lrecl.;
end;

stop;
run;
```

### log

```
echo IN_FILE=C:\program-files\SASHome\SASFoundation\9.?\sasv9.cfg
NOTE: The infile "C:\program-files\SASHome\SASFoundation\9.?\sasv9.cfg" is:
      Filename=C:\program-files\SASHome\SASFoundation\9.?\sasv9.cfg,
      -config "C:\program-files\SASHome\SASFoundation\9.?\nls\en\sasv9.cfg"
```

Program 4 echos the main, the largest, configuration file.

#### Program 4 echo-config-main.sas

```
%let in_file = %sysfunc(sysget(sascfg))\sasv9.cfg;
%include 'echo-any-file.sas';
```

---

#### log

```
echo IN_FILE= C:\program-files\SASHome\SASFoundation\9.?\nls\en\sasv9.cfg
...
-SET SASROOT "C:\program-files\SASHome\SASFoundation\9.?"
...
-SET SASAUTOS ("!SASROOT\core\sasmacro"
...
                "!SASROOT\graph\sasmacro"
                "!SASROOT\stat\sasmacro"
            )
```

---

**notes:** The environment variable `sasroot` is used in the environment variable `sasautos`, which is the list of directories of the `fileref` `sasautos`, which is the default value of option `sasautos`.

Note that the list of directories in `sasautos` is enclosed in parentheses.

---

## sql procedure, dictionary.options

### overview

This is the overview, which consists of a list of topics in this section.

- describe dictionary options
  - freq optstart
- 

Program 5 writes the data structure of sql dictionary.options to the log.

#### Program 5 proc-sql-describe-dictionary-options.sas

```
proc sql; describe table dictionary.options;
quit;
```

---

#### log

```
create table dictionary.options
  optname char(32) label='Option Name',
  opttype char(8) label='Option type',
  offset num label='Offset into option value',
  setting char(1024) label='Option Setting',
  optdesc char(160) label='Option Description',
  level char(8) label='Option Location',
  optstart char(8) label='Option Set',
  group char(32) label='Option Group'
```

---

### freq optstart

Program 6 was used to produce table 2, shown above on pg. 3.

#### Program 6 proc freq, tables optstart

```
proc sql; create table dictionary_options_optstart as
  select optstart
  from dictionary.options;
quit;
proc freq data = &syslast;
  tables optstart;
run;
```

---

## Setup for testing

### overview

This is the overview, which consists of a list of topics in this section. This section shows how to create a suite of files for a unit test of a user-written configuration file.

- functions used here
- write test suite, config
- sas.cmd
- project sasv9.cfg
- templates:
  - autoexec.sas
  - initstmt.sas
  - termstmt.sas
- 01-config-test.\*
- 02-autoexec-test.\*

### functions used here

---

These functions are used in the following programs.

- `getoption(option-name<,options>)`                      fetch value of option
  - `putn(value,format)`    echo computed value
  - `%scan(string,n,delimiter)`    fetch word of string
  - `%sysfunc(fn(...))`    access to data-step functions
  - `%sysget(evar)`    fetch value of environment variable (evar)
-

## write-test-suite- config.sas

Program 7 writes to the specified folder a set of files for this article.

### Program 7 write test suite of config test files

```
*name: write-test-suite-config.sas;
%let folder=c:\temp\sas-test;
%let folder=..\sas-startup-config-test-suite;
data _null_;
    file "&folder\sas.cmd";
put    'rem name: sas.cmd '
    /   '"" "%sysget(sasroot)\sas.exe" "" %*';

    file "&folder\01-config-test.bat";
put    'rem name: 01-config-test.bat'
    /   'sas      01-config-test -verbose -pagesize max';

    file "&folder\01-config-test.sas";
put    '*name:    01-config-test.sas;';

    file "&folder\sasv9.cfg";
put    '/*name: sasv9.cfg */'
    /   "-sasinitialfolder '.'"
    /   "-set site_inc 'C:\SAS-projects\SAS-site\sas-includes' "
    /   "-set site_inc '.'"
    /   '-initstmt "" "%include '!site_inc\initstmt.sas';" ""'
    /   '-termstmt "" "%include '!site_inc\termstmt.sas';" ""';

    file "&folder\02-autoexec-test.bat";
put    'rem name: 02-autoexec-test.bat'
    /   'sas 02-autoexec-test -echoauto -source -source2';

    file "&folder\02-autoexec-test.sas";
put    '*name: 02-autoexec-test.sas;'
    /   'options pagesize=max;'
    /   'filename _all_ list;'
    /   'libname _all_ list;'
    /   '%put echo mautosource=%sysfunc(getoption(mautosource));'
    /   '%put %sysfunc(getoption(sasautos,keyword));'
    /   '%let mvar2=text.2 for show sort order;'
    /   '%let mvar1=text.1 for termstmt;';

    file "&folder\autoexec.sas";
put    '*name: autoexec.sas;'
    /   "title1 '???' new project ???';"
    /   "filename project '.';"
    /   "libname library '.';";

    file "&folder\initstmt.sas";
put    '*name: initstmt.sas;'
    /   '%put echo %sysfunc(getoption(initstmt));';

    file "&folder\termstmt.sas";
put    '*name: termstmt.sas;'
    /   '%put echo %sysfunc(getoption(termstmt,keyexpand));';

stop;
run;
```

---



Program 8 centralizes the call to `sas.exe`.

#### Program 8 project `sas.cmd`

```
rem name: sas.cmd
"C:\...\SASHome\SASFoundation\9.?\sas.exe" %*
```

**notes:** Percent asterisk (`%*`) is the DOS syntax for 'pass all command-line parameters to program'.  
DOS strings are numbered: `%1 ... %n`; asterisk means 'all'.

---

#### `sasv9.cfg`

Program 9 replicates the syntax of the main configuration file by providing an environment variable `site_inc` which is then used in options `initstmt` and `termstmt`.

#### Program 9 project `sasv9.cfg`

```
/*name: sasv9.cfg */
-sasinitialfolder '.'
-set site_inc 'C:\SAS-projects\SAS-site\sas-includes'
-set site_inc '.'
-initstmt "%include '!site_inc\initstmt.sas';"
-termstmt "%include '!site_inc\termstmt.sas';"
```

**notes:** Option `sasinitialfolder` is the argument to Display Manager command: File Save As; dot means 'here': current folder.  
Environment variable `site_inc` has two values as a reminder that you may, in the future, wish to provide access to subroutine files `initstmt.sas` and `termstmt.sas` for all projects in a central folder.  
The arguments of both contain a reference to the environment variable `!site_inc`;  
macro variables are referenced with an ampersand (`&`);  
environment variable are referenced with a bang: exclamation point (`!`).

---

#### templates:

##### `autoexec.sas`

```
*name: autoexec.sas;
title1 '??? new project ???';
filename project '.';
libname library '.';
```

##### `initstmt.sas`

```
*name: initstmt.sas;
%put echo %sysfunc(getoption(initstmt));
```

##### `termstmt.sas`

```
*name: termstmt.sas;
%put echo %sysfunc(getoption(termstmt,keyexpand));
```

**notes:** Function `getoption` has a required option of `option-name` and an optional argument of `keyexpand` which is used to format the value; both options contain an environment variable reference: `!site_inc`, which will be resolved when echoed.

---

## unit test: sasv9.cfg

### overview

This is the overview, which consists of a list of topics in this section.

- 01-config-test.\*
- sections in 01-config-test
- 01-config-test.log, pg. 1, 2, 3, 4, 12

### 01-config-test.\*

Files 01-config-test.bat and 01-config-test.sas in program listing 10 are used for a unit test of the configuration file sasv9.cfg.

#### Program 10 01-config-test.\*

```
rem name: 01-config-test.bat
sas      01-config-test -verbose -pagesize max

*name:   01-config-test.sas;
```

**notes:** Option `-verbose` writes the settings of the system options to the log. Option `-pagesize max` removes page breaks from the log, which may be 500+ lines; at 50 lines/page, more than 10 pages.

### sections in 01-config-test

- Options **specified** in the **config files**:
- Options set internally at initialization:
- Options **specified** on the **command line**:
- Options set internally at initialization:
- Options set internally during session startup:
- Options set from locale:
- Options set internally at initialization:

### 01-config-test.log, pg. 1

```
23 Options specified in the config file
24     C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg:
25 -----
26 CONFIG = "C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"
27 -----
```

**notes:** line 26, primary config file:  
compare to listing from program echo-config-main.sas, pg. 6;

```
28 Options specified in the config file
29     C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg:
30 -----
31 ...
32 SET = SASROOT "C:\program-files\SASHome\SASFoundation\9.4"
33 JREOPTIONS = (-DPFS_TEMPLATE=!SASROOT\tkjava\sasmisc\qrfstpt.xml
34
35 ...
46 SET = SASAUTOS ("!SASROOT\core\sasmacro"
47 ...
48             "!SASROOT\graph\sasmacro"
49             "!SASROOT\stat\sasmacro"
50             )
```

**notes:** Lines 32–33 show the config file allocating an environment variable, `sasroot` and using it in an option, `jreoptions`; lines 46–50 show its use in another environment variable, `sasautos`, which is the the list of directories for the *fileref* and option `sasautos`. Note the list of directories is enclosed in parentheses.

01-config-test.log,  
pg 2

```
88 SET = MYSASFILES "?FOLDERID_Documents\My SAS Files\9.4"  
89 SASUSER = "?FOLDERID_Documents\My SAS Files\9.4"  
90 WORK = "!TEMP\SAS Temporary Files"  
91 ...  
92 SET = SASCFG "C:\program-files\SASHome\SASFoundation\9.4\nls\en"
```

**notes:** ?FOLDERID\_Documents

88–89: question mark (?) is special character used to reference an API

92: !TEMP is a reference to a Windows environment variable

! → In previous versions of SAS software, environment variable MYSASFILES and option SASUSER fetched values of Windows environment variables named CSIDL\*

01-config-test.log,  
pg 3

```
...  
111 -----  
112 Options specified in the config file  
113 C:\SAS-projects\2023-config\sas-startup-config-test-suite\sasv9.cfg:  
114 -----  
115 SASINITIALFOLDER = '.',  
116 SET = site_inc 'C:\SAS-projects\SAS-site\sas-includes'  
117 SET = site_inc '.',  
118 INITSTMT = "%include '!site_inc\initstmt.sas';"  
119 TERMSTMT = "%include '!site_inc\termstmt.sas';"  
120 -----  
121 Options set internally at initialization:  
122 -----  
123 CONFIG = ( "C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg"  
124 "C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg"  
125 "C:\SAS-projects\2023-config\sas-startup-config-test-suite\sasv9.cfg" )  
126 -----  
127 Options specified on the command line:  
128 -----  
129 SYSIN = 01-config-test  
130 VERBOSE  
131 PAGESIZE = max  
132 -----  
133 Options set internally at initialization:  
134 -----  
135 AUTOEXEC = "C:\SAS-projects\2023-config\sas-startup-config-test-suite\autoexec.sas"
```

**notes:** Options set internally at initialization:

123–125: the list of configuration files and order in which they are 'loaded',

! → note the use of the SAS HelpDesk verb *loaded*:  
this means the last environment variable allocation  
or option assignment is the value used

Options specified on the command line:

129–131: command-line assignments 'loaded' after configuration file assignments

Options set internally at initialization:

135: option autoexec is set to value of sasinitialfolder  
plus default filename: autoexec.sas

**01-config-test.log,  
pg. 4**

```
155 NOTE: Copyright (c) 2002-2012 by SAS Institute Inc., Cary, NC, USA.
156 NOTE: SAS (r) Proprietary Software 9.?
157 ...
158 NOTE: SAS initialization used:
159     real time          2.55 seconds
160
161 ===== Processed Configuration File(s) =====
162 C:\program-files\SASHome\SASFoundation\9.4\sasv9.cfg
163 C:\program-files\SASHome\SASFoundation\9.4\nls\en\sasv9.cfg
164 C:\SAS-projects\2023-config\sas-startup-config-test-suite\sasv9.cfg
165
166 ===== Environment Variable Options =====
167 <none>
168
169 Option      Value
170 =====
171 AUTOEXEC    C:\SAS-projects\2023-config\sas-startup-config-test-suite\autoexec.sas
172 ...
```

---

**notes:** beginning of notes from option sysin=01-config-test.sas  
Processed Configuration File(s)  
162–164: compare to 01-config-test.log, page 3, lines 123–125, above

---

**01-config-test.log,  
pg. 12**

```
550 NOTE: AUTOEXEC processing beginning; file is
551     C:\SAS-projects\2023-config\sas-startup-config-test-suite\autoexec.sas.
552
553 NOTE: Libref LIBRARY was successfully assigned as follows:
554     Engine:          V9
555     Physical Name: C:\SAS-projects\2023-config\sas-startup-config-test-suite
556
557 NOTE: AUTOEXEC processing completed.
558
559 echo %include '!site_inc\initstmt.sas'
560 1          *name: 01-config-test.sas;
561
562 echo TERMSTMT=%include '\termstmt.sas'
563 NOTE: SAS Institute Inc., SAS Campus Drive, Cary, NC USA 27513-2414
564 NOTE: The SAS System used:
565     real time          2.60 seconds
566     cpu time           0.46 seconds
```

---

**notes:**  
550–557 option autoexec processing is  
%include 'autoexec.sas'\nosource2;  
there is no echo of the filename statement  
but any *libref(s)* allocated are duly noted  
559 compare to termstmt, line 562  
560 line 1 from sysin  
562 compare to initstmt, line 559: keyexpand produces TERMSTMT=

The next section shows polishing of %put echo for options initstmt and termstmt shown above on pg. 9.

---

## template for initstmt and termstmt

### overview

This section explains how to use the start-up only options `initstmt`, which statements are executed after the `autoexec`, and `termstmt`, which statements are executed at `endsas`. The task here is to replicate the beginning ... completed notes like those in `autoexec` processing. This is the list of topics in this section.

- `autoexec` notes
- syntax
- `initstmt.sas` and log
- `termstmt.sas` and log

### autoexec notes

---

```
NOTE: AUTOEXEC processing beginning; file is  
      C:\SAS-projects\2023-config\sas-startup-config\autoexec.sas.  
...  
NOTE: AUTOEXEC processing completed.
```

---

### syntax

- semicolon :** Task 1 is how to write a note with a semicolon in it. This is accomplished by enclosing the semicolon and CR/LF in the `%str()` function:  
`%str(  
; )`
- %scan :** Task 2 is parsing the value of the options, which contains two words. The `initstmt` value is `'%include "file-specification";'`. The `%scan(string,n,delimiters)` function is used to pick the 2nd word; the `%sysfunc(getoption(option-name))` pair of functions is used to fetch the value of the option.
- job info :** Task 3 is to write a set of notes containing job information: startup date+time-stamp, userid, and job-name; this information is necessary to match the log with the program when parsing logs of program runs.
- time :** Task 4: The read-only macro variable `sysprocessid` is a 32-character hexadecimal string, the first 16 characters of which contain the program startup date+time-stamp.
-

Program 11 is the replacement for program `initstmt.sas` on page 9.

#### Program 11 `initstmt.sas`

```
* name initstmt.sas;
%put Note initstmt processing beginning%str(
)file is %scan(%sysfunc(getoption(initstmt)),2,%str( ));
%let _string = JOB-INFO:;
%put &_string datetime_hex16=%substr(&sysprocessid,1,16);
%put &_string datetime=%sysfunc
      (putn(%substr(&sysprocessid,1,16)x,datetime21.2));
%put &_string &=sysuserid;
%put &_string %sysfunc(getoption(sysin,keyword)) #;
%symdel _string;
%put Note initstmt processing completed.;
```

log

---

```
Note: initstmt processing beginning; file is '!site_inc\initstmt.sas'.
JOB-INFO: datetime_hex16=40D5944000000000
JOB-INFO: datetime=31APR2023:11:23:58.13
JOB-INFO: SYSUSERID=Ronald
JOB-INFO: SYSIN=C:\SAS-projects\2023-config\02-autoexec-test.sas #
Note: initstmt processing completed
```

---

Program 11 is the replacement for program `initstmt.sas` on page 9.

#### Program 12 `termstmt.sas`

```
* name: termstmt.sas;
%put Note: termstmt processing beginning%str(
)file is %scan(%sysfunc(getoption(termstmt,keyexpand)),2,%str( ));
%put echo list global _user_ macro variables, if any;
%put _user_;
%put Note: termstmt processing completed.;
```

log

---

```
Note: termstmt processing beginning; file is '!\termstmt.sas'.
echo list global _user_ mvars, if any
Note: termstmt processing completed
```

---

## unit test: autoexec.sas

### overview

This section shows the programs and log.

- 02-autoexec-test.\*
- 02-autoexec-test.log, initstmt
- 02-autoexec-test.log, program
- 02-autoexec-test.log, termstmt

---

Program 13 shows the .bat and .sas files used to test an autoexec.sas file.

### Program 13 02-autoexec-test: .bat and .sas

```
rem name: 02-autoexec-test.bat
sas 02-autoexec-test -echoauto -source -source2

*name: 02-autoexec-test.sas;
options pagesize=max;
filename _all_ list;
libname _all_ list;
%put echo mautosource=%sysfunc(getoption(mautosource));
%put %sysfunc(getoption(sasautos,keyword));
%let mvar2=text.2 for show sort order;
%let mvar1=text.1 for termstmt;
```

---

**notes:** Options `-echoauto` and `-source` turn on the display of statements in the autoexec file.

Option `-source2` turns on the display of statements of `%include` files; see files `initstmt.sas` and `termstmt.sas`.

---

### 02-autoexec-test.log, initstmt

```
NOTE: %INCLUDE (level 1) file !site_inc\initstmt.sas is file
      C:\SAS-projects\2023-config\sas-startup-config\initstmt.sas.
1      ** name: initstmt.sas;
2      +%put Note: initstmt processing beginning%str(
3      +)file is %scan(%sysfunc(getoption(initstmt)),2,%str( ));
Note: initstmt processing beginning; file is '!site_inc\initstmt.sas'
4      +
5      +%let _string = JOB-INFO;
6      +%put &_string: datetime_hex16=%substr(&sysprocessid,1,16);
JOB-INFO: datetime_hex16=41DCF8DA18B73B64
7      +%put &_string: datetime=%sysfunc
8      + (putn(%substr(&sysprocessid,1,16)x,datetime21.2));
JOB-INFO: datetime=31APR2023:06:40:34.00
9      +%put &_string: &=sysuserid;
JOB-INFO: SYSUSERID=Ronald
10     +%put &_string: %sysfunc(getoption(sysin,keyword)) #;
JOB-INFO: SYSIN=C:\SAS-projects\2023-config\sas-startup-config\02-autoexec-test.sas #
11     +%symdel _string;
12     +
13     +%put Note: initstmt processing completed.;
Note: initstmt processing completed.
NOTE: %INCLUDE (level 1) ending.
```

---

## 02-autoexec-test.log, program

```
1          *name: 02-autoexec-test.sas;
2          options pagesize=max;
3          filename _all_ list;
NOTE: Fileref= PROJECT
      Physical Name= C:\SAS-projects\2023-config\sas-startup-configuration
4          libname _all_ list;
NOTE: Libref= LIBRARY
      Scope=      Program 02-autoexec-test
      Engine=     V9
      Physical Name= C:\SAS-projects\2023-config\sas-startup-configuration
      Filename= C:\SAS-projects\2023-config\sas-startup-configuration
...
5          %put echo mautosource=%sysfunc(getoption(mautosource));
echo mautosource=MAUTOSOURCE
6          %put %sysfunc(getoption(sasautos,keyword));
SASAUTOS=SASAUTOS
7          %let mvar2=text.2 for show sort order;
8          %let mvar1=text.1 for termstmt;
```

---

**notes:** Options `mautosource` and `sasautos` are the pair of options of the *autocall* of macros.

## 02-autoexec-test.log, termstmt

```
NOTE: %INCLUDE (level 1) file !site_inc\termstmt.sas is file
      C:\SAS-projects\2023-config\sas-startup-configuration\termstmt.sas.
2          +* name: termstmt.sas;
3          +%put Note: termstmt processing beginning%str(
4          +)file is %scan(%sysfunc(getoption(termstmt,keyexpand)),2,%str( ));
Note: termstmt processing beginning; file is '\termstmt.sas'
5          +%put echo list global _user_ macro variables, if any;
echo list global _user_ macro variables, if any
6          +%put _user_;
GLOBAL MVAR1 text.1 for termstmt
GLOBAL MVAR2 text.2 for show sort order
7          +%put Note: termstmt processing completed.;
Note: termstmt processing completed.
NOTE: %INCLUDE (level 1) ending.
```

---



## Summary

### Suggested Reading

- api: Wikipedia editors, et al., *API: application programming interface*
- Microsoft Corp: csidl: Microsoft Corp., *CSIDL: constant special item ID list*; known-folder-id: Microsoft Corp., *KnownFolderId: ?FolderId\_Documents*
- SAS Institute: csidl: SAS Institute, *CSIDL: Configuring SAS for DCOM*; files used by SAS: SAS Institute, *Files Used by SAS :: SAS(R) 9.4 Companion for Windows, Fifth Edition*
- companions: Fehd, "An Autoexec Companion, Allocating Location Names during Startup" autoexec companion  
Fehd, "A Sysparm Companion, Passing Values to a Program from the Command Line" sysparm companion
- testing: Fehd, "Writing Testing-Aware Programs that Self-Report when Testing Options are True" writing testing-aware programs using command-line options `echoauto` and `verbose`
- sql: Fehd, "How To Use proc SQL select into for List Processing" using sql for list processing;  
see program `ProcSQL-D-Macros-select-into-list-ordered.sas` for use in `termstmt.sas`

### Conclusion

---

Configuration files contain two kinds of statements: allocations of environment variables and options' assignments. SAS configuration files use environment variables as arguments for options. Environment variables contain list of directories, libraries, with files that contain its functions, procedures, and macros. Writing configuration files can be a simple matter. The purpose of this paper is to provide a suite of tools that make testing user-written configuration files easier.

---

### Author Information

Ronald J. Fehd  
LinkedIn  
affiliation  
also known as

Ron.Fehd.macro.maven at gmail dot com  
<https://www.linkedin.com/in/ronald-fehd-5125991/>  
Fragile-Free Software Institute  
macro maven on SAS-L

---

### Trademarks

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. In the USA and other countries ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

---

## References

- Fehd, Ronald J. (2007). "Writing Testing-Aware Programs that Self-Report when Testing Options are True". In: *NorthEast SAS Users Group Conference Proceedings*. Coders' Corner, 20 pp.; topics: options used while testing: echoauto, mprint, source2, verbose; variable testing in data step or macros; call execute; references. URL: <http://www.lexjansen.com/nesug/nesug07/cc/cc12.pdf>.
- (2010). "How To Use proc SQL select into for List Processing". In: *SouthEast SAS Users Group Conference Proceedings*. Hands On Workshop, 40 pp.; topics: writing constant text, and macro calls, using macro %do loops; references. URL: <http://analytics.ncsu.edu/sesug/2010/H0W06.Fehd.pdf>.
- (2018a). "A Sysparm Companion, Passing Values to a Program from the Command Line". In: *South-East SAS Users Group Conference Proceedings*. 8 pp.; shows use of sysparm as macro variable and option which can be assigned value on command line in batch programs; program parse-sysparm parses a list of comma-separated values (csv) of form var1=value1,var2=value2,...,varN=valueN into macro variables. URL: [http://www.lexjansen.com/sesug/2018/SESUG2018\\_Paper-197\\_Final\\_PDF.pdf](http://www.lexjansen.com/sesug/2018/SESUG2018_Paper-197_Final_PDF.pdf).
- (2018b). "An Autoexec Companion, Allocating Location Names during Startup". In: *SouthEast SAS Users Group Conference Proceedings*. 16 pp.; autocall macros, global symbol table, catrefs, filerefs, librefs, cexist catalogs, exist data set, sasautos. URL: [http://www.lexjansen.com/sesug/2018/SESUG2018\\_Paper-196\\_Final\\_PDF.pdf](http://www.lexjansen.com/sesug/2018/SESUG2018_Paper-196_Final_PDF.pdf).
- Microsoft Corp. (2018). *CSIDL: constant special item ID list*. URL: <https://docs.microsoft.com/en-us/windows/win32/shell/csidl>.
- (2021). *KnownFolderId: ?FolderId\_Documents*. URL: <https://docs.microsoft.com/en-us/windows/desktop/shell/knownfolderid>.
- SAS Institute (2021a). *CSIDL: Configuring SAS for DCOM*. URL: [https://support.sas.com/rnd/itech/doc9/admin\\_oma/sasserver/comdcom/sascfg.html](https://support.sas.com/rnd/itech/doc9/admin_oma/sasserver/comdcom/sascfg.html).
- (2021b). *Files Used by SAS :: SAS(R) 9.4 Companion for Windows, Fifth Edition*. URL: <http://support.sas.com/documentation/cdl/en/hostwin/69955/HTML/default/viewer.htm#p0bmj7wjme32ayn1h4wim7trkhp6.htm>.
- Wikipedia editors, et al. (2018). *API: application programming interface*. URL: <https://en.wikipedia.org/wiki/API>.
-